#!/bin/zsh

### Print where antidote is cloning bundles.
#
# usage: antidote home [-h|--help]
#
# Can be overridden by setting `$ANTIDOTE_HOME`.
#
#function antidote-home {
  emulate -L zsh; setopt local_options $_adote_funcopts

  typeset -g REPLY=
  local o_help
  zparseopts $_adote_zparopt_flags -- h=o_help -help=h || return 1

  if (( $#o_help )); then
    antidote-help home
    return
  fi

  local result
  if [[ -n "$ANTIDOTE_HOME" ]]; then
    result=$ANTIDOTE_HOME
  else
    result=$(__antidote_get_cachedir "antidote")
  fi
  print $result
  typeset -g REPLY=$result
#}
